F07UWF (CTPTRI/ZTPTRI) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

F07UWF (CTPTRI/ZTPTRI) computes the inverse of a complex triangular matrix, using packed storage.

2 Specification

```
SUBROUTINE F07UWF(UPLO, DIAG, N, AP, INFO)
ENTRY      ctptri(UPLO, DIAG, N, AP, INFO)
INTEGER            N, INFO
complex            AP(*)
CHARACTER*1        UPLO, DIAG
```

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine forms the inverse of a complex triangular matrix A using packed storage. Note that the inverse of an upper (lower) triangular matrix is also upper (lower) triangular.

4 References

[1] Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

5 Parameters

1: UPLO — CHARACTER*1 *Input*

On entry: indicates whether A is upper or lower triangular as follows:

if UPLO = 'U', then A is upper triangular;
if UPLO = 'L', then A is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: DIAG — CHARACTER*1 *Input*

On entry: indicates whether A is a non-unit or unit triangular matrix as follows:

if DIAG = 'N', then A is a non-unit triangular matrix;
if DIAG = 'U', then A is a unit triangular matrix; the diagonal elements are not referenced and are assumed to be 1.

Constraint: DIAG = 'N' or 'U'.

3: N — INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

4: AP(∗) — ***complex*** array *Input/Output*

Note: the dimension of the array AP must be at least max(1,N∗(N+1)/2).

On entry: the n by n triangular matrix A, packed by columns. More precisely, if UPLO = 'U', the upper triangle of A must be stored with element a_{ij} in AP$(i+j(j-1)/2)$ for $i \leq j$;if UPLO = 'L', the lower triangle of A must be stored with element a_{ij} in AP$(i+(2n-j)(j-1)/2)$ for $i \geq j$. If DIAG = 'U', the diagonal elements of the matrix are not referenced and are assumed to be 1;the same storage scheme is used whether DIAG = 'N' or 'U'.

On exit: A is overwritten by A^{-1}, using the same storage format as described above.

5: INFO — INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, a_{ii} is zero and the matrix A is singular.

7 Accuracy

The computed inverse X satisfies

$$|XA - I| \leq c(n)\epsilon|X||A|,$$

where $c(n)$ is a modest linear function of n, and ϵ is the ***machine precision***.

Note that a similar bound for $|AX - I|$ cannot be guaranteed, although it is almost always satisfied.

The computed inverse satisfies the forward error bound

$$|X - A^{-1}| \leq c(n)\epsilon|A^{-1}||A||X|.$$

See Du Croz and Higham [1].

8 Further Comments

The total number of real floating-point operations is approximately $\frac{4}{3}n^3$.

The real analogue of this routine is F07UJF (STPTRI/DTPTRI).

9 Example

To compute the inverse of the matrix A, where

$$A = \begin{pmatrix} 4.78+4.56i & 0.00+0.00i & 0.00+0.00i & 0.00+0.00i \\ 2.00-0.30i & -4.11+1.25i & 0.00+0.00i & 0.00+0.00i \\ 2.89-1.34i & 2.36-4.25i & 4.15+0.80i & 0.00+0.00i \\ -1.89+1.15i & 0.04-3.69i & -0.02+0.46i & 0.33-0.26i \end{pmatrix},$$

using packed storage.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*     F07UWF Example Program Text
*     Mark 15 Release. NAG Copyright 1991.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          NMAX
      PARAMETER        (NMAX=8)
      CHARACTER        DIAG
      PARAMETER        (DIAG='N')
*     .. Local Scalars ..
      INTEGER          I, IFAIL, INFO, J, N
      CHARACTER        UPLO
*     .. Local Arrays ..
      complex          AP(NMAX*(NMAX+1)/2)
      CHARACTER        CLABS(1), RLABS(1)
*     .. External Subroutines ..
      EXTERNAL         ctptri, X04DDF
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F07UWF Example Program Results'
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) N
      IF (N.LE.NMAX) THEN
*
*        Read A from data file
*
         READ (NIN,*) UPLO
         IF (UPLO.EQ.'U') THEN
            READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
         ELSE IF (UPLO.EQ.'L') THEN
            READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
         END IF
*
*        Compute inverse of A
*
         CALL ctptri(UPLO,DIAG,N,AP,INFO)
*
*        Print inverse
*
         WRITE (NOUT,*)
         IFAIL = 0
         CALL X04DDF(UPLO,DIAG,N,AP,'Bracketed','F7.4','Inverse',
     +               'Integer',RLABS,'Integer',CLABS,80,0,IFAIL)
      END IF
      STOP
*
      END
```

9.2 Program Data

```
F07UWF Example Program Data
  4                                                          :Value of N
  'L'                                                        :Value of UPLO
 ( 4.78, 4.56)
 ( 2.00,-0.30) (-4.11, 1.25)
 ( 2.89,-1.34) ( 2.36,-4.25) ( 4.15, 0.80)
 (-1.89, 1.15) ( 0.04,-3.69) (-0.02, 0.46) ( 0.33,-0.26)  :End of matrix A
```

9.3 Program Results

```
F07UWF Example Program Results

Inverse
                     1                 2                 3                 4
1  ( 0.1095,-0.1045)
2  ( 0.0582,-0.0411) (-0.2227,-0.0677)
3  ( 0.0032, 0.1905) ( 0.1538,-0.2192) ( 0.2323,-0.0448)
4  ( 0.7602, 0.2814) ( 1.6184,-1.4346) ( 0.1289,-0.2250) ( 1.8697, 1.4731)
```